
NAG Library Routine Document

F08ZTF (ZGGRQF)

Note: before using this routine, please read the Users' Note for your implementation to check the interpretation of bold italicised terms
and other implementation-dependent details.

1 Purpose

F08ZTF (ZGGRQF) computes a generalized RQ factorization of a complex matrix pair A;Bð Þ, where
A is an m by n matrix and B is a p by n matrix.

2 Specification

SUBROUTINE F08ZTF (M, P, N, A, LDA, TAUA, B, LDB, TAUB, WORK, LWORK,
INFO)

&

INTEGER M, P, N, LDA, LDB, LWORK, INFO
COMPLEX (KIND=nag_wp) A(LDA,*), TAUA(min(M,N)), B(LDB,*),

TAUB(min(P,N)), WORK(max(1,LWORK))
&

The routine may be called by its LAPACK name zggrqf.

3 Description

F08ZTF (ZGGRQF) forms the generalized RQ factorization of an m by n matrix A and a p by n matrix
B

A ¼ RQ; B ¼ ZTQ;

where Q is an n by n unitary matrix, Z is a p by p unitary matrix and R and T are of the form

R ¼

��n�m m

m 0 R12
; if m � n;

1
CA

0
B@

n

m� n R11

n R21

; if m > n;

8>>>>>>>>>><
>>>>>>>>>>:

with R12 or R21 upper triangular,

T ¼

�� n

n T11
p� n 0

; if p � n;

�� p n� p

p T11 T12 ; if p < n;

8>>>><
>>>>:

with T11 upper triangular.

In particular, if B is square and nonsingular, the generalized RQ factorization of A and B implicitly
gives the RQ factorization of AB�1 as

AB�1 ¼ RT�1
� �

ZH:

5 Arguments

1: M – INTEGER Input

On entry: m, the number of rows of the matrix A.

Constraint: M � 0.

2: P – INTEGER Input

On entry: p, the number of rows of the matrix B.

Constraint: P � 0.

3: N – INTEGER Input

On entry: n, the number of columns of the matrices A and B.

Constraint: N � 0.

4: AðLDA; �Þ – COMPLEX (KIND=nag_wp) array Input/Output

Note: the second dimension of the array A must be at least max 1;Nð Þ.
On entry: the m by n matrix A.

On exit: if m � n, the upper triangle of the subarray Að1 : m;n�mþ 1 : nÞ contains the m by
m upper triangular matrix R12.

If m � n, the elements on and above the m� nð Þth subdiagonal contain the m by n upper
trapezoidal matrix R; the remaining elements, with the array TAUA, represent the unitary matrix
Q as a product of min m;nð Þ elementary reflectors (see Section 3.3.6 in the F08 Chapter
Introduction).

5: LDA – INTEGER Input

On entry: the first dimension of the array A as declared in the (sub)program from which F08ZTF
(ZGGRQF) is called.

Constraint: LDA � max 1;Mð Þ.

6: TAUAðmin M;Nð ÞÞ – COMPLEX (KIND=nag_wp) array Output

On exit: the scalar factors of the elementary reflectors which represent the unitary matrix Q.

7: BðLDB; �Þ – COMPLEX (KIND=nag_wp) array Input/Output

Note: the second dimension of the array B must be at least max 1;Nð Þ.
On entry: the p by n matrix B.

On exit: the elements on and above the diagonal of the array contain the min p; nð Þ by n upper
trapezoidal matrix T (T is upper triangular if p � n); the elements below the diagonal, with the
array TAUB, represent the unitary matrix Z as a product of elementary reflectors (see
Section 3.3.6 in the F08 Chapter Introduction).

8: LDB – INTEGER Input

On entry: the first dimension of the array B as declared in the (sub)program from which F08ZTF
(ZGGRQF) is called.

Constraint: LDB � max 1; Pð Þ.

9: TAUBðmin P;Nð ÞÞ – COMPLEX (KIND=nag_wp) array Output

On exit: the scalar factors of the elementary reflectors which represent the unitary matrix Z.

10: WORKðmax 1;LWORKð ÞÞ – COMPLEX (KIND=nag_wp) array Workspace

On exit: if INFO ¼ 0, the real part of WORKð1Þ contains the minimum value of LWORK
required for optimal performance.

11: LWORK – INTEGER Input

On entry: the dimension of the array WORK as declared in the (sub)program from which
F08ZTF (ZGGRQF) is called.

If LWORK ¼ �1, a workspace query is assumed; the routine only calculates the optimal size of
the WORK array, returns this value as the first entry of the WORK array, and no error message
related to LWORK is issued.

Suggested value: for optimal performance, LWORK � max N;M; Pð Þ �max nb1; nb2;nb3ð Þ,
where nb1 is the optimal block size for the RQ factorization of an m by n matrix by
F08CVF (ZGERQF), nb2 is the optimal block size for the QR factorization of a p by n matrix by
F08ASF (ZGEQRF), and nb3 is the optimal block size for a call of F08CXF (ZUNMRQ).

Constraint: LWORK � max 1;N;M; Pð Þ or LWORK ¼ �1.

12: INFO – INTEGER Output

On exit: INFO ¼ 0 unless the routine detects an error (see Section 6).

6 Error Indicators and Warnings

INFO < 0

If INFO ¼ �i, argument i had an illegal value. An explanatory message is output, and execution
of the program is terminated.

7 Accuracy

The computed generalized RQ factorization is the exact factorization for nearby matrices Aþ Eð Þ and
Bþ Fð Þ, where

Ek k2 ¼ O � Ak k2 and Fk k2 ¼ O � Bk k2;
and � is the machine precision.

8 Parallelism and Performance

F08ZTF (ZGGRQF) is threaded by NAG for parallel execution in multithreaded implementations of the
NAG Library.

F08ZTF (ZGGRQF) makes calls to BLAS and/or LAPACK routines, which may be threaded within the
vendor library used by this implementation. Consult the documentation for the vendor library for further
information.

Please consult the X06 Chapter Introduction for information on how to control and interrogate the
OpenMP environment used within this routine. Please also consult the Users' Note for your
implementation for any additional implementation-specific information.

9 Further Comments

The unitary matrices Q and Z may be formed explicitly by calls to F08CWF (ZUNGRQ) and F08ATF
(ZUNGQR) respectively. F08CXF (ZUNMRQ) may be used to multiply Q by another matrix and
F08AUF (ZUNMQR) may be used to multiply Z by another matrix.

The real analogue of this routine is F08ZFF (DGGRQF).

10 Example

This example solves the least squares problem

minimize
x

c�Axk k2 subject to Bx ¼ d

where

A ¼

0:96� 0:81i �0:03þ 0:96i �0:91þ 2:06i �0:05þ 0:41i
�0:98þ 1:98i �1:20þ 0:19i �0:66þ 0:42i �0:81þ 0:56i
0:62� 0:46i 1:01þ 0:02i 0:63� 0:17i �1:11þ 0:60i
0:37þ 0:38i 0:19� 0:54i �0:98� 0:36i 0:22� 0:20i
0:83þ 0:51i 0:20þ 0:01i �0:17� 0:46i 1:47þ 1:59i
1:08� 0:28i 0:20� 0:12i �0:07þ 1:23i 0:26þ 0:26i

0
BBBBB@

1
CCCCCA
;

B ¼ 1 0 �1 0
0 1 0 �1

� �
; c ¼

�2:54þ 0:09i
1:65� 2:26i

�2:11� 3:96i
1:82þ 3:30i

�6:41þ 3:77i
2:07þ 0:66i

0
BBBBB@

1
CCCCCA

and d ¼ 0
0

� �
:

The constraints Bx ¼ d correspond to x1 ¼ x3 and x2 ¼ x4.

The solution is obtained by first obtaining a generalized RQ factorization of the matrix pair A;Bð Þ. The
example illustrates the general solution process, although the above data corresponds to a simple
weighted least squares problem.

Note that the block size (NB) of 64 assumed in this example is not realistic for such a small problem,
but should be suitable for large problems.

10.1 Program Text

Program f08ztfe

! F08ZTF Example Program Text

! Mark 26 Release. NAG Copyright 2016.

! .. Use Statements ..
Use nag_library, Only: dznrm2, nag_wp, zgemv, zggrqf, ztrmv, ztrtrs, &

zunmqr, zunmrq
! .. Implicit None Statement ..

Implicit None
! .. Parameters ..

Complex (Kind=nag_wp), Parameter :: one = (1.0E0_nag_wp,0.0E0_nag_wp)
Integer, Parameter :: nb = 64, nin = 5, nout = 6

! .. Local Scalars ..
Real (Kind=nag_wp) :: rnorm

Integer :: i, info, lda, ldb, lwork, m, n, p
! .. Local Arrays ..

Complex (Kind=nag_wp), Allocatable :: a(:,:), b(:,:), c(:), d(:), &
taua(:), taub(:), work(:), x(:)

! .. Intrinsic Procedures ..
Intrinsic :: min

! .. Executable Statements ..
Write (nout,*) ’F08ZTF Example Program Results’
Write (nout,*)

! Skip heading in data file
Read (nin,*)
Read (nin,*) m, n, p
lda = m
ldb = p
lwork = nb*(p+n)
Allocate (a(lda,n),b(ldb,n),c(m),d(p),taua(n),taub(n),work(lwork),x(n))

! Read A, B, C and D from data file

Read (nin,*)(a(i,1:n),i=1,m)
Read (nin,*)(b(i,1:n),i=1,p)
Read (nin,*) c(1:m)
Read (nin,*) d(1:p)

! Compute the generalized RQ factorization of (B,A) as
! B = (0 T12)*Q, A = Z*(R11 R12)*Q, where T12 and R11
! ( 0 R22)
! are upper triangular
! The NAG name equivalent of zggrqf is f08ztf

Call zggrqf(p,m,n,b,ldb,taub,a,lda,taua,work,lwork,info)

! Set Qx = y. The problem then reduces to:
! minimize (Ry - Z^Hc) subject to Ty = d
! Update c = Z^H*c -> minimize (Ry-c)
! The NAG name equivalent of zunmqr is f08auf

Call zunmqr(’Left’,’Conjugate transpose’,m,1,min(m,n),a,lda,taua,c,m, &
work,lwork,info)

! Putting y = (y1), solve T12*w = d for w, storing result in d
! (w )
! The NAG name equivalent of ztrtrs is f07tsf

Call ztrtrs(’Upper’,’No transpose’,’Non-unit’,p,1,b(1,n-p+1),ldb,d,p, &
info)

If (info>0) Then
Write (nout,*) ’The upper triangular factor of B is singular, ’
Write (nout,*) ’the least squares solution could not be computed’
Go To 100

End If

! From first n-p rows of (Ry-c) we have
! R11*y1 + R12*w = c(1:n-p) = c1
! Form c1 = c1 - R12*w = R11*y1

! The NAG name equivalent of zgemv is f06saf
Call zgemv(’No transpose’,n-p,p,-one,a(1,n-p+1),lda,d,1,one,c,1)

! Solve R11*y1 = c1 for y1, storing result in c(1:n-p)
! The NAG name equivalent of ztrtrs is f07tsf

Call ztrtrs(’Upper’,’No transpose’,’Non-unit’,n-p,1,a,lda,c,n-p,info)

If (info>0) Then
Write (nout,*) ’The upper triangular factor of A is singular, ’
Write (nout,*) ’the least squares solution could not be computed’
Go To 100

End If

! Copy y into x (first y1, then w)
x(1:n-p) = c(1:n-p)
x(n-p+1:n) = d(1:p)

F08 – Least-squares and Eigenvalue Problems (LAPACK) F08ZTF

Mark 26 F08ZTF.5



! Compute x = (Q**H)*y
! The NAG name equivalent of zunmrq is f08cxf

Call zunmrq(’Left’,’Conjugate transpose’,n,1,p,b,ldb,taub,x,n,work, &
lwork,info)

! The least squares solution is in x, the remainder here is to compute
! the residual, which equals c2 - R22*w.

! Upper triangular part of R22 first
! The NAG name equivalent of ztrmv is f06sff

Call ztrmv(’Upper’,’No transpose’,’Non-unit’,min(m,n)-n+p, &
a(n-p+1,n-p+1),lda,d,1)

Do i = 1, min(m,n) - n + p
c(n-p+i) = c(n-p+i) - d(i)

End Do

If (m<n) Then

! Additional rectangular part of R22
! The NAG name equivalent of zgemv is f06saf

Call zgemv(’No transpose’,m-n+p,n-m,-one,a(n-p+1,m+1),lda,d(m-n+p+1), &
1,one,c(n-p+1),1)

End If

! Compute norm of residual sum of squares.
! The NAG name equivalent of dznrm2 is f06jjf

rnorm = dznrm2(m-(n-p),c(n-p+1),1)

! Print least squares solution x
Write (nout,*) ’Constrained least squares solution’
Write (nout,99999) x(1:n)

! Print estimate of the square root of the residual sum of
! squares

Write (nout,*)
Write (nout,*) ’Square root of the residual sum of squares’
Write (nout,99998) rnorm

100 Continue

99999 Format (4(’ (’,F7.4,’,’,F7.4,’)’,:))
99998 Format (1X,1P,E10.2)

End Program f08ztfe

10.2 Program Data

F08ZTF Example Program Data

6 4 2 : m, n and p

( 0.96,-0.81) (-0.03, 0.96) (-0.91, 2.06) (-0.05, 0.41)
(-0.98, 1.98) (-1.20, 0.19) (-0.66, 0.42) (-0.81, 0.56)
( 0.62,-0.46) ( 1.01, 0.02) ( 0.63,-0.17) (-1.11, 0.60)
( 0.37, 0.38) ( 0.19,-0.54) (-0.98,-0.36) ( 0.22,-0.20)
( 0.83, 0.51) ( 0.20, 0.01) (-0.17,-0.46) ( 1.47, 1.59)
( 1.08,-0.28) ( 0.20,-0.12) (-0.07, 1.23) ( 0.26, 0.26) : A

( 1.00, 0.00) ( 0.00, 0.00) (-1.00, 0.00) ( 0.00, 0.00)
( 0.00, 0.00) ( 1.00, 0.00) ( 0.00, 0.00) (-1.00, 0.00) : B

(-2.54, 0.09)
( 1.65,-2.26)
(-2.11,-3.96)
( 1.82, 3.30)
(-6.41, 3.77)
( 2.07, 0.66) : c

( 0.00, 0.00)
( 0.00, 0.00) : d

F08ZTF NAG Library Manual

F08ZTF.6 Mark 26



10.3 Program Results

F08ZTF Example Program Results

Constrained least squares solution
( 1.0874,-1.9621) (-0.7409, 3.7297) ( 1.0874,-1.9621) (-0.7409, 3.7297)

Square root of the residual sum of squares
1.59E-01

